
1. Generic programming

What is datatype-generic programming?

I Programming with the structure of types

I Conversion functions map user datatypes to/from
representation types

I Generic functions are defined on representation types

Generic functions work for all types for which we can write
conversion functions.

Generic representation I

Haskell supports the definition of algebraic datatypes, like:

data Logic = Logic ∨ Logic -- disjunction
| Var String -- variables
| Not Logic -- negation

To represent these, we need to know how to handle:

I Different alternatives: disjoint sums.

I Arguments of a constructor: products.

I Constructors and field labels.

I Primitive types: String , Int , . . .

I . . .

Generic representation II

We need to translate every datatype to this set of constructs
and apply the appropriate code in the right place.

Haskell’s data construct combines several features: type
abstraction, type recursion, (labeled) sums, and (possibly
labeled) products, but they are essentially sums of products.

To represent them we can use the following representation
datatypes:

data (f + g) r = L (f r) | R (g r) -- Choice
data (f × g) r = f r × g r -- Multiple arguments
data K a r = K a -- Constants
data I r = I r -- Recursive occurrences
data U r = U -- No arguments

Generic representation III

We also need to represent constructors:

data C c f r = C (f r)

class Con c where
conName :: t c (f :: ∗ → ∗) r → String

We encapsulate conversion to and from the generic
representation using a type class. The generic type is given
using a type family:

type family PF a :: ∗ → ∗
class Regular a where

from :: a → PF a a
to :: PF a a → a

Generic representation IV

Back to our Logic example:

type instance PF Logic = (I × I ) -- disjunction
+ (K String) -- variables
+ I -- negation

instance Regular Logic where
from (p ∨ q) = L ((I p) × (I q))
from (Var x ) = R (L (K x ))
from (Not p) = R (R (I p))

to (L ((I p) × (I q))) = p ∨ q
to (R (L (K x ))) = Var x
to (R (R (I p))) = Not p

We omit constructor information for simplicity.

Generic functions: gmap

Now we can write generic functions:

class GMap f where
gmap :: (a → b)→ f a → f b

instance GMap I where
gmap f (I r) = I (f r)

instance GMap (K a) where
gmap (K x ) = K x

instance GMap U where
gmap U = U

instance (GMap f ,GMap g)⇒ GMap (f + g) where
gmap f (L x ) = L (gmap f x )
gmap f (R x ) = R (gmap f x )

instance (GMap f ,GMap g)⇒ GMap (f × g) where
gmap f (x × y) = gmap f x × gmap f y

Generic functions: gshow I

Another function we can define is generic show. For that we
need to use constructor information.

class GShow f where
gshowf :: (a → String)→ f a → String

instance GShow I where
gshowf f (I r) = f r

instance (Show a)⇒ GShow (K a) where
gshowf (K x ) = show x

instance GShow U where
gshowf U = ""

instance (Con c,GShow f )⇒ GShow (C c f ) where
gshowf f cx@(C x ) = "(" ++ conName cx ++ " "

++ gshowf f x ++ ")"

Generic functions: gshow II

instance (GShow f ,GShow g)⇒ GShow (f + g) where
gshowf f (L x ) = gshowf f x
gshowf f (R x ) = gshowf f x

instance (GShow f ,GShow g)⇒ GShow (f × g) where
gshowf f (x × y) = gshowf f x ++ " " ++ gshowf f y

This function works only on the generic representations. For
normal datatypes we first have to convert them:

gshow :: (Regular a,GShow (PF a))⇒ a → String
gshow x = gshowf gshow (from x )

At the recursive occurrences we apply gshow again.

2. Optimizing generics through inlining

Efficiency I

While representation types are useful, they incur a performance
penalty:

I Generic functions keep converting back and forth

I Generic representation types are present in the final
generated code

I Even “fast” generic programming libraries typically perform
2–4 times slower than handwritten variants

I “Slower” libraries can be up to 8–16 times slower

Efficiency II

Generic representation types should not be present in the
generated code. Generic functions can be specialized to
particular types.

We can see that if we inline definitions and apply equational
reasoning we can remove the generic representations.

As an example, let us see one-level generic identity on the Logic
datatype:

gidLogic :: Logic → Logic
gidLogic = to ◦ gmap id ◦ from

Efficiency III

to (gmap id (from l))

⇒ { choose l to be p ∨ q (other constructors similar) }
to (gmap id (from (p ∨ q)))

≡ { definition of fromLogic }
to (gmap id (L (I p × I q)))

≡ { definition of gmap+, gmap× }
to (L (gmap id (I p) × gmap id (I q)))

≡ { definition of gmapI }
to (L (I (id p) × (I (id q))))

≡ { definition of id , toLogic }
p ∨ q

Core code I

Can we not get the compiler to do the same for us? The core
code GHC generates for our example

gidLogic :: Logic → Logic
gidLogic = to ◦ gmap id ◦ from

is

gidO1
Logic :: Logic → Logic

gidO1
Logic = λ(x :: Logic)→ to (from x )

This is good, but not ideal. We also know that
toLogic ◦ fromLogic ≡ id .

Core code II

The problem is that the compiler is conservative with
inlining—replacing function calls with their body. We can force
inlining by tweaking some flags:

Flag Default Abbr.

-funfolding-creation-threshold 45 CT
-funfolding-use-threshold 6 UT

Compiling with -O2 -funfolding-use-threshold=60
produces the wanted result:

gidO2UT60
Logic :: Logic → Logic

gidO2UT60
Logic = λ(x :: Logic)→ x

Core code III

For gshow , with standard optimizations we get:

gshowO1
Logic :: Logic → String

gshowO1
Logic = λ(x :: Logic)→

case (from x ) ‘cast ‘ (sym (trans . . .)) of w {
L y → . . .
R y → . . .}

But we can force inlining to obtain a better result:

gshowCT90UT30
Logic :: Logic → String

gshowCT90UT30
Logic = λ(x :: Logic)→ case x of w {

(∨) p q → (++) . . . gshowCT90UT30
Logic p . . . gshowCT90UT30

Logic q . . .

Var v → (++) . . . show v . . .

Not p → (++) . . . gshowCT90UT30
Logic p . . .

Const b → (++) . . . show b . . .}

3. A benchmark suite for generics

A benchmark suite for generics: functions

To visualize the impact of increased inlining we designed a
benchmark suite of generic programs. We will show two
functions:

show Requires constructor information, such as name
and fixity.

update Transform all odd Int values by adding one to
them, or prepend all non-empty String values
with a "y".

In our paper we present also the results for generic equality,
map and read.

A benchmark suite for generics: datatypes

We use two datatypes. The Tree datatype is a simple labeled
binary leaf tree:

data Tree a = Bin a (Tree a) (Tree a) | Leaf

The Logic type is similar to the one we introduced before, only
with more constructors:

data Logic = Impl Logic Logic | Equiv Logic Logic
| Conj Logic Logic | Disj Logic Logic
| Not Logic | Var String | T | F

A benchmark suite for generics: libraries

We have chosen a few representative, mainstream, and
maintained libraries to benchmark:

emgm Extensible and Modular Generics for the Masses.
Its fundamental characteristic is to encode
datatype representations through a type class.

syb Scrap Your Boilerplate is a very popular library
based on generic combinators and type-safe cast.
It comes with GHC.

regular The library described in the introduction.

multirec The first approach able to express mutually
recursive datatypes. Structurally similar to
regular, but makes use of a few more advanced
concepts to deal with mutual recursion.

Results: show for Tree

Results: show for Logic

Results: update for Tree

Results: update for Logic

4. Conclusions and future work

Conclusions

I Generic programs do not have to be slow

I Inlining is the way to go

I Facilities for inlining are already present in the compiler
and can be reused for optimizing generics

I Both emgm and regular are fast and can be optimized to
handwritten code speed with inlining

I The slowest (but most popular) generic programming
library is syb

I multirec is not benefiting much from increased inlining,
as opposed to the similar regular library

Future work

I Specifying the behavior of the inliner should be more
localized: use the INLINE pragmas of the upcoming
version of GHC

I Not all libraries benefit equally from increased inlining:
why?

I Are GADTs preventing inlining in multirec?
I What can we do about syb?

I Investigate generic producers more thoroughly

